

Arrays: Random Circles

Let's use arrays to make a program that draws random circles each time we click the mouse.

Step One: Draw One Random Circle

It's always a good idea to make a simpler version of an idea first, and then make it more complex. Let's create a program that will generate a random circle.

```
float x;
float y;
float circleSize;
color circleColor;

void setup() {
  size(600, 600);
  x = random(width);
  y = random(height);
  circleSize = random(10, 100);
  circleColor = color(random(255), random(255), random(255));
}

void draw() {
  background(0);
  fill(circleColor);
  circle(x, y, circleSize);
}
```

Each time we run this program, it will generate a random circle at a random location on the screen, and with a random color. Great!

Step Two: Click for New Circle

Now, let's refactor a bit, to put our code in methods. Then, we can use generate a new circle each time we click the mouse, using Processing's `mousePressed()` method.

```

float x;
float y;
float circleSize;
color circleColor;

void setup() {
  size(600, 600);
  newCircle();
}

void draw() {
  background(0);
  showCircle();
}

void newCircle() {
  x = random(width);
  y = random(height);
  circleSize = random(10, 100);
  circleColor = color(random(255), random(255), random(255));
}

void showCircle() {
  fill(circleColor);
  circle(x, y, circleSize);
}

void mousePressed() {
  newCircle();
}

```

Step Three: Add Arrays

Now let's convert our single variables to arrays. Then, we will need to add a loops to our methods that create and show the circles, to repeat the code for each item in the array. It is also traditional (and logical) to rename our variables and method to plural forms, to show that they are arrays with many values. So **circleSize** becomes **circleSizes**, and **showCircle** becomes **showCircles**.

```

float[] circleXs = new float[10];
float[] circleYs = new float[10];
float[] circleSizes = new float[10];
color[] circleColors = new color[10];

void setup() {
  size(600, 600);
  newCircles();
}

void draw() {
  background(0);
  showCircles();
}

void newCircles() {
  for (int i = 0; i < circleXs.length; i++) {
    circleXs[i] = random(width);
    circleYs[i] = random(height);
    circleSizes[i] = random(10, 100);
    circleColors[i] = color(random(255), random(255), random(255));
  }
}

void showCircles() {
  for (int i = 0; i < circleXs.length; i++) {
    fill(circleColors[i]);
    circle(circleXs[i], circleYs[i], circleSizes[i]);
  }
}

void mousePressed() {
  newCircles();
}

```

Challenges

- Add a variable for the number of balls. Use this variable to control the size of all of the arrays.
- Can you modify the program so that you can add additional circles each time you click?